

you-try-it-01.xlsx
Step-by-Step Guide
ver. 7/26/2016

Abstract

This document provides step-by-step instructions for the Excel workbook you-try-it-01.xlsx (saved from Excel 2010). The worksheets contain data for practice exercises keyed to Chapter 1 of:

Basics of Analytical Chemistry and Chemical Equilibria by Brian M. Tissue
(John Wiley: New York, 2013).

<u>Worksheets in the workbook</u>	<u>page in this guide</u>
1.A conversions.....	2
1.B precision.....	6
1.C outlier.....	11
1.D LOD-LOQ.....	15
1.E calibration	19
1.F standard-addition.....	23

General Advice

It is often useful to work out the first exercise of each worksheet on paper. After obtaining a result, compare your result to the answer in the worksheet. Next try writing formulas to do the calculation(s). If you do not get the same answer, try writing formulas stepwise and check intermediate steps to error check your work. Try to write formulas to be general so that you can copy them to use for multiple cases.

Version History

6/10/2009	First posting.
3/1/2010	Worksheets reorganized to match Chapter 1 revisions.
7/26/2016	Formatting revisions.

For updates visit the text support website:

<http://www.achem.org>

1.A conversions

This worksheet contains measurement results in common units. The exercise is to convert these results to other units.

Concepts	Concept Synopsis
unit conversions	Having only one universal concentration unit would be nice, but you'll see many different units to express results in convenient numerical ranges.
algebra	When using a conversion factor, we are really just multiplying a value by one – the numerator and denominator of a conversion factor are equivalent.

1. Table 1.A.1 provides several examples of physical measurements in common units. Express each case in Table 1.A.1 in the other units that are listed. Some conversion factors are below the table.

Any conversion calculation is a bookkeeping exercise. Although this first case is a simple change in the units in the numerator, I will show the units explicitly for both the numerator and denominator. This approach is useful for more complicated conversions, such as ones that we'll do later. An equivalent expression for 25.00 mph is:

$$25.00 \frac{\text{mi}}{1.000 \text{ h}}$$

In the conversion of this measurement to km/h only the units in the numerator change. You can find conversion factors in many reference sources and online. Typing "mi in km" in the search box at <http://www.google.com/> and clicking Search returns:

$$1 \text{ mi} = 1.609344 \text{ kilometers}$$

Using this equality in our calculation will cancel the mi units, leaving us with the result in km/h:

$$25.00 \frac{\cancel{\text{mi}}}{1.000 \text{ h}} \times \frac{1.6093 \text{ km}}{1.000 \cancel{\text{mi}}} = 40.23 \frac{\text{km}}{1.000 \text{ h}}$$

Google also provides calculator functions. You can check calculations by entering them in the search box, e.g., "25 mph in km/h" returns:

$$25 \text{ mph} = 40.2336 \text{ km/h}$$

The km/mi conversion factor is in cell D26, so in cell H21 I enter the formula:

$$= F21*\$D\$26$$

I use absolute referencing with dollar signs, i.e., $\$D\26 , for the conversion factor in case I want to copy and reuse this formula. The last thing that I do is adjust the cell formatting in cell H21 to report the result with the correct number of significant figures: 40.23 km/h. I follow a similar procedure for the other conversions. The next figure shows how the worksheet looks after writing formulas for the other cells.

	A	B	C	D	E	F	G	H	I	J	K
18	Table 1.A.1										
19											
20	sample	measurement			value	units	Convert to:				
21	vehicle	velocity			25.00	mph	40.23	km/h	11.18	m/s	
22	10 % sucrose solution	density			1.034	g/cm ³	1.034	g/mL	1034	kg/m ³	
23	7Ag85Cu8Sn brazing alloy	density			4.80	oz/cu in	8.30	g/mL	8304	kg/m ³	
24											
25				value	units	value	units	value	units	value	units
26				1.6093	km/mi	1000	g/kg	1	mL/cm ³	10000	ppm/%
27				28.3495	g/oz	100	cm/m	1000	mL/L	1000	ppm/‰
28				16.3871	mL/cu in	1000	mm/m	1000	L/m ³		

Because Excel has no capacity to keep track of units, conversions are often less error-prone when written on paper first. Excel is most useful when you have a large number of repetitive calculations. In those cases it is worth the effort to write and validate formulas on paper and then to enter them in a worksheet to complete multiple calculations simultaneously.

- Table 1.A.2 provides a list of analytical results with common units. Express each case in Table 1.A.2 in the other units that are listed. The density of each solution is given in column E.

The general principle of converting concentration units is the same as for any other numerical data. Two common cases in chemistry are converting between mass and volume using density and between mass and moles using formula weight. Since the first case in Table 1.A.2 has a density of 1.00 g/mL, which is typical for dilute solutions, I'll use the conversion of 30.0 ‰ NaCl in seawater as my example here. 30.0 parts per thousand NaCl is equivalent to:

$$30.0 \frac{\text{g NaCl}}{1000 \text{ g soln}}$$

where “soln” is a generic abbreviation for the sample solution. To convert to ppm, which is the number of grams of substance per 1×10^6 g of sample, we must convert the denominator from ‘1000 g soln’ to ‘ 1×10^6 g of soln’. This is done by multiplying the denominator by 1000. As in any algebraic conversion, we must not alter the value, so we also multiply the numerator by 1000. Our conversion factor is then (1000/1000):

$$30.0 \frac{\text{g NaCl}}{1000 \text{ g soln}} \times \frac{1000}{1000} = 3.00 \times 10^4 \frac{\text{g NaCl}}{1.000 \times 10^6 \text{ g soln}} = 3.00 \times 10^4 \text{ ppm NaCl}$$

This result illustrates why we use different units in different types of analyses, 30.0 ‰ is simply a more convenient scale for salinity than is 3.00×10^4 ppm. In Excel the formula is simply:

$$=F36*\$J\$27$$

where cell J27 has the conversion factor of 1000 ppm/‰.

Now converting 30.0 ‰ NaCl to M (molarity, mol/L) will illustrate the use of density and formula weight. Again we set up the conversion on paper to check that units cancel appropriately:

$$30.0 \frac{\text{g NaCl}}{1000 \text{ g soln}} \times \frac{1.000 \text{ mol NaCl}}{58.443 \text{ g NaCl}} \times \frac{1.025 \text{ g soln}}{1.000 \text{ mL soln}} \times \frac{1000 \text{ mL}}{1.000 \text{ L}} = \frac{0.526 \text{ mol NaCl}}{1.000 \text{ L soln}}$$

Note that g NaCl and g soln cancel to leave us with mol NaCl per mL soln. Converting mL to L gives us our desired result of 0.526 M NaCl. For the calculation to convert the 30.0 ‰ value, in cell J36 I enter:

$$=(F36/ \$F\$43)*\$E\$36$$

where cell F43 contains the NaCl formula weight and cell E36 has the solution density. Note that the factor of 1000 to convert mL to L cancels with the 1000 g soln for the ‰ units and I did not include these values in the formula. The last step is to format the result to 3 significant figures:

	A	B	C	D	E	F	G	H	I	J	K
32	Table 1.A.2										
33											
34	analysis		density (g/mL)	value	units	Convert to:					
35	nitrate in water		1.000	1.9	ppm		wt %				M
36	NaCl in seawater		1.025	30.0	‰	3.00E+04	ppm	0.526			M
37	acetic acid in vinegar		1.010	5.5	% by wt		ppm				M
38	ethanol in beverage		1.020	4.2	% (v/v)		wt %				M
39											
40				species	formula	wt		density			
41				NO ₃ ⁻	62.005	g/mol					
42				NaCl	58.443	g/mol					
43				CH ₃ COOH	60.052	g/mol					
44				C ₂ H ₅ OH	46.068	g/mol	0.789	g/mL			

Some extra explanation is useful for the last case in Table 1.A.2. Percent by volume, often denoted with (v/v), is the volume of a substance per 100 mL of solution. 70 % ethanol in water is a common lab disinfectant. To make this solution, add 70 mL of ethanol to a 100 mL container and then add water to bring the total volume of solution to 100 mL. Mixing 70 mL of ethanol and 30 mL of water will not give you the same result. The co-mixing of the water and ethanol molecules gives you slightly less than 100 mL of solution.

For the last case in Table 1.A.2, 4.2 % ethanol can be expressed as:

$$4.2 \frac{\text{mL ethanol}}{100 \text{ mL soln}}$$

Given this starting point, we can use the density of ethanol to convert the numerator to g of ethanol:

$$4.2 \frac{\text{mL ethanol}}{100 \text{ mL soln}} \times \frac{0.789 \text{ g ethanol}}{1.00 \text{ mL ethanol}} = 3.31 \frac{\text{g ethanol}}{100 \text{ mL soln}}$$

This calculation shows that multiplying by the solute density converts ‘percent by volume’ to ‘percent by weight’.

With the value for the density of ethanol being in cell H44, the Excel formula that I enter in cell H38 is:

$$=F38*\$H\$44$$

The rest of this conversion to convert to molarity is similar to the other calculations above. In cell 38 I enter:

$$=H38*10/F44$$

The overall results are shown in the next figure.

	A	B	C	D	E	F	G	H	I	J	K
32	Table 1.A.2										
33											
34	analysis		density (g/mL)	value	units	Convert to:					
35	nitrate in water		1.000	1.9	ppm	1.9E-04	wt %	3.1E-05	M		
36	NaCl in seawater		1.025	30.0	‰	3.00E+04	ppm	0.526	M		
37	acetic acid in vinegar		1.010	5.5	% by wt	5.5E+04	ppm	0.92	M		
38	ethanol in beverage		1.020	4.2	% (v/v)	3.3	wt %	0.72	M		

1.B precision

This worksheet provides two tables of repetitive measurements to calculate the mean, standard deviation, and %-RSD.

Concepts	Concept Synopsis
precision	The repeatability or reproducibility in a series of identical measurements or analytical results.
descriptive statistics	There are a number of different measures of precision that are useful in different situations. The most common measure is the standard deviation, s . Others include percentage relative standard deviation, %-RSD, standard error, s_m , and confidence intervals. See Chapter 1 for descriptions and equations to calculate these measures of precision.

1. Calculate the wt-% for each measurement in Table 1.B.1.
Find the mean, standard deviation, and %-RSD.

Click the mouse in cell E18 and type:

$$=(D18/C18)*100$$

This formula will calculate the weight percent for the first measurement (Trial 1). Copy this formula to cells E19:E21 to repeat the calculation for the other measurements. In cells E23, E24, and E25 enter formulas to determine the mean, standard deviation, and %-RSD for the measurements:

$$E23: =AVERAGE(E18:E21)$$

$$E24: =STDEV(E18:E21)$$

$$E25: =(E24/E23)*100$$

The results are shown in the next figure, where I've also added the column title "wt-%" and explanatory labels and units in adjacent columns:

	A	B	C	D	E	F	G
15	Table 1.B.1 Protein analysis of extra crunchy peanut butter.						
16							
17		Trial	Sample wt (g)	Protein wt (g)	wt-%		
18		1	9.91	1.92	19.37		
19		2	10.17	2.04	20.06		
20		3	10.54	2.09	19.83		
21		4	10.01	2.05	20.48		
22							
23				mean:	19.94	%	
24				std dev:	0.46	%	
25				relative standard deviation:	2.3	%-RSD	

I purposely used data in the example above that were percentage measurements to point out the difference between the units of the data, in this case "%", and "%-RSD". This coincidence can often be confusing so it is best to always use %-RSD or RSD when referring to relative standard deviation.

2. Column E contains formulas for statistical results for up to 20 data points in Table 1.B.2.

Write formulas to calculate the variance and standard error.

Use the table of *t* values to the right to find the 90 and 99 % confidence levels.

Verify your results using Data -> Data Analysis -> Descriptive Statistics.

Table 1.B.2 is set up to calculate descriptive statistics of up to twenty data points entered in column C. Only five data points are given in this exercise, but we will write formulas so that data points can be added or deleted. Several formulas are already entered in column E, giving the results in the next figure. I have formatted the results to three decimal places to be able to compare values. In reporting a result, I would reduce this to two decimal places to match the number of significant figures of the data.

	A	B	C	D	E	F	G
28	Table 1.B.2 Descriptive Statistics						
29							
30		Trial	Data		Statistics		
31		1	3.44	N =	5		
32		2	3.11	sum =	15.830		
33		3	2.98	mean =	3.166		
34		4	3.27	std dev =	0.188		
35		5	3.03				
36		6		%-RSD =			
37		7		variance =			
38		8		std error =			
39		9					t values
40		10		90-% C.L. =			
41		11		95-% C.L. =	0.234		2.776
42		12		99-% C.L. =			

Now to write formulas for the other listed measures of precision, the %-RSD is the fraction of the standard deviation divided by the mean multiplied by 100 %. In cell E36 I enter:

$$=(\$/E\$34/\$/E\$33)*100$$

The variance is the square of the standard deviation, so in cell E37 I enter:

$$=\$/E\$34^2$$

The standard error is the standard deviation divided by the square root of N , so in cell E38 I enter:

$$=\$/E\$34/SQRT(\$/E\$31)$$

The expression for confidence limit, $C.L.$, is:

$$C.L. = \frac{t s}{\sqrt{N}}$$

where N is the number of data points, s is the standard deviation, and the value of t is taken from a table and depends on $N-1$ and the desired level of confidence. The simplest approach to calculate $C.L.$ is to write a formula and type in the appropriate t value. The disadvantage of typing a numerical value in a formula is that you might not remember to change the value if the data set changes. A better approach is to place the value of t in a separate cell and use that cell location in your formula. For the value of t in cell G41, the formula for the 95-% $C.L.$ in cell E41 is:

$$=\$/G\$41*\$/E\$34/SQRT(\$/E\$31)$$

I must still remember to change the *t* value in cell G41 if the data set changes. Another way to place the appropriate *t* value in cell G41 is to use the VLOOKUP function to find the *t* value in a table of values. Cell G41 contains the formula:

=VLOOKUP(\$E\$31-1,\$L\$5:\$R\$21,4,FALSE)

The parameters in the VLOOKUP function are:

=VLOOKUP(lookup_value, table_array, col_index_num, [range_lookup])

Parameter	This Example	Explanation
lookup_value	\$E\$31-1 (<i>N</i> -1)	The value that the function locates in the first column of the table.
table_array	\$L\$5:\$R\$21	The worksheet array that contains the table of <i>t</i> values.
col_index_num	4	The column that contains the value to be returned. The fourth column for the 95-% <i>C.L.</i>
[range_lookup]	FALSE	An option to not return an approximate match, i.e., to return only an exact matches.

To enter formulas to calculate the 90-% and 99-% confidence levels, I copy cell E41 to cells E40 and E42 and cell G41 to cells G40 and G42. The VLOOKUP formulas must be revised to return the correct *t* value. The *t* values for the 90-% and 99-% *C.L.* are in the third and sixth columns, respectively. I revise the col_index_num to be:

G40: =VLOOKUP(\$E\$31-1,\$L\$5:\$R\$21,3,FALSE)

G42: =VLOOKUP(\$E\$31-1,\$L\$5:\$R\$21,6,FALSE)

The results now look like the following figure:

	A	B	C	D	E	F	G
28		Table 1.B.2 Descriptive Statistics					
29							
30		Trial	Data		Statistics		
31		1	3.44	N =	5		
32		2	3.11	sum =	15.830		
33		3	2.98	mean =	3.166		
34		4	3.27	std dev =	0.188		
35		5	3.03				
36		6		%-RSD =	5.954		
37		7		variance =	0.036		
38		8		std error =	0.084		
39		9					<i>t</i> values
40		10		90-% C.L. =	0.180		2.132
41		11		95-% C.L. =	0.234		2.776
42		12		99-% C.L. =	0.388		4.604

The last part of this exercise is to use Excel’s built-in descriptive statistics. These are accessed from Data Analysis on the Data ribbon. After finding Descriptive Statistics in the list and clicking OK, enter the following in the dialog box (see figure below):

Input Range: C31:C50
 Output Range: I30 or any empty cell
 Summary statistics: checked
 Confidence Level for Mean: 99 %

The results will look like the figure below. You can repeat this analysis for different confidence levels by changing the “Confidence Level for Mean” value in the dialog box. A disadvantage of this built-in analysis is that the values returned by Descriptive Statistics will not change if the data changes. If you edit the data, you will need to run Descriptive Statistics again.

<i>Column1</i>	
Mean	3.166
Standard Error	0.08429709
Median	3.11
Mode	#N/A
Standard Deviation	0.18849403
Sample Variance	0.03553
Kurtosis	-0.76940565
Skewness	0.78202913
Range	0.46
Minimum	2.98
Maximum	3.44
Sum	15.83
Count	5
Confidence Level(99.0%)	0.38811182

1.C outlier

This worksheet contains a set of replicate data that contains an apparent outlier. The exercise is to use statistical tests to determine if the apparent outlier may be discarded from the set.

Concepts	Concept Synopsis
outlier	An outlier is a data point that is not representative of the true value due to a measurement error or other factor.
discarding an outlier	When no error is known to have occurred, but a measurement appears to be very different from other values, statistical tests can validate discarding the outlier. Two common tests are Dixon's Q-test and Peirce's criterion.

1. Determine which data value in Table 1.C.1 is a potential outlier and calculate Q . Determine if the outlier may be discarded at the 95-% or 99-% confidence levels.

The procedure for Dixon's Q -test is to calculate the value Q for a suspected outlier and compare it to a critical value, Q_c . Values of Q_c are taken from a table and depend on N and the desired confidence level. The next figure shows the data in columns B and C, basic statistics of the data, and values of Q_c for $N = 5$. The 36.5 mV measurement appears much lower than the other values in the data set. The calculations in this worksheet will test if this value may be discarded.

	A	B	C	D	E	F	G
12		Table 1.C.1 Calcium Potentiometry					
13		Trial	ISE Result (mV)		Results		
14		1	39.8	N =	5	$Q_c (95\%) = 0.710$	
15		2	36.5	sum =	195.0	$Q_c (99\%) = 0.821$	
16		3	39.9	mean =	39.00		
17		4	39.2	std dev =	1.42		
18		5	39.6				

The expression for Dixon's Q-test is:

$$Q = \frac{|\text{outlier} - \text{closest}|}{|\text{max} - \text{min}|}$$

where the values are:

- outlier: the suspected outlier
- closest: the data value closest to the suspected outlier
- max: the largest value in the data set
- min: the smallest value in the data set

The value of Q for the 36.5 mV result is:

$$Q = \frac{|\text{outlier} - \text{closest}|}{|\text{max} - \text{min}|} = \frac{|36.5 - 39.2|}{|39.9 - 36.5|} = 0.794$$

On comparing the calculated Q of 0.794 to the Q_c values, the 36.5 mV outlier may be discarded at the 95-% confidence level ($Q_c = 0.710$), but not at the 99-% confidence level ($Q_c = 0.821$).

There are two useful Excel functions to automate finding values based on relative size:

=SMALL(range,k)

=LARGE(range,k)

The SMALL function returns the lowest values in the cell range, in this case C14:C18. The numerical argument, k, indicates the smallest, 1, the next smallest, 2, and so on. The LARGE function operates in the same way, finding the largest values in the specified cell range. We will use these functions to automate finding potential outliers, i.e., the smallest and largest values in a data set. In cells C23 and C24 I enter:

=SMALL(\$C\$14:\$C\$18,1)

=LARGE(\$C\$14:\$C\$18,1)

To easily identify the data value farthest from the mean, I enter in cells D23 and D24 the formulas:

=C23-\$E\$16

=C24-\$E\$16

These differences are called the deviations or residuals from the mean. In this particular example the smallest value of 36.5 is 2.5 from the mean and the largest value of 39.9 is 0.9 from the mean. The question we will answer is whether or not the 36.5 value is so

far from the mean that there was likely an error in the measurement. To calculate the Dixon Q parameter, I find the closest values to the min and max values by entering in cells E23 and E24:

```
=SMALL($C$14:$C$18,2)
=LARGE($C$14:$C$18,2)
```

Although I am only testing for the minimum value of 36.5 mV, I will write formulas for both the min and max so that I can reuse these formulas for other data sets. Now typing the Q calculation in cells F23 and F24:

```
=ABS(C23-E23)/ABS(C24-C23)
=ABS(C24-E24)/ABS(C24-C23)
```

where the ABS() function returns the absolute value of the value in the parentheses. I can compare the calculated Q to the Q_c values using an IF function. In cells G23 and G24 I enter, respectively:

```
= IF($F23>$G$14,"discard","do not discard")
= IF($F24>$G$14,"discard","do not discard")
```

The IF function compares the two values, and if the logical relationship is true displays “discard”, and if not true displays “do not discard”. The overall results are shown in the next figure:

	A	B	C	D	E	F	G	H
21		Dixon Q -test calculation						
22			value	deviation	closest value	Dixon's Q	Q_c (95 %)	Q_c (99 %)
23		min =	36.5	-2.5	39.2	0.794	discard	do not discard
24		max =	39.9	0.9	39.8	0.029	do not discard	do not discard

2. Repeat using Peirce's criterion.

The procedure to use Peirce’s criterion is to first calculate the deviation from the mean for a suspected outlier:

$$d = | x_i - \bar{x} |$$

where x_i , is the suspected outlier and \bar{x} is the mean. This deviation is then compared to the maximum deviation, d_{\max} , that is expected for a random distribution given the number of data values. d_{\max} is found from:

$$d_{\max} = sR$$

where s is the sample standard deviation and R is taken from a table of values. If the deviation of the suspected outlier is larger than d_{\max} , it is improbable that the outlier occurred by chance. The suspected outlier may therefore be discarded if this condition is met:

$$|x_i - \bar{x}| > d_{\max}$$

In cells C30:D31 I enter formulas to find the min, max, and their deviations from the mean:

C30: =SMALL(\$C\$14:\$C\$18,1)

C31: =LARGE(\$C\$14:\$C\$18,1)

D30: =ABS(C30-\$E\$16)

D31: =ABS(C31-\$E\$16)

where I also include the ABS() function to find the absolute value of the deviations. As before, I do calculations for both the min and max to be able to reuse the formulas for other data sets. The value of R from the adjacent table is placed in cells E30 and E31 using a look-up function:

=VLOOKUP(\$E\$14,\$K\$30:\$L\$44,2,FALSE)

I enter in cells F30 and F31 the maximum deviation, $d_{\max} = sR$, using:

F30: =E30*\$E\$17

F31: =E31*\$E\$17

In cells G30 and G31 I enter an IF function to test if $|x_i - \bar{x}| > d_{\max}$:

G30: =IF(D30>F30,"discard","do not discard")

G31: =IF(D31>F31,"discard","do not discard")

The results are shown in the following figure. The 36.5 mV value may be discarded based on Peirce's criterion.

	A	B	C	D	E	F	G
28	Peirce's criterion						
29			value	deviation	R	d_{\max}	result
30	min =		36.5	2.5	1.51	2.15	discard
31	max =		39.9	0.9	1.51	2.15	do not discard

1.D LOD-LOQ

This worksheet contains two data sets for calculating Limit of Detection (LOD) and Limit of Quantitation (LOQ).

Concepts	Concept Synopsis
Limit of Detection (LOD)	The minimum concentration at which an analyte may be reported as being detected. Generally taken to be the concentration that produces an analytical signal three times greater than the noise level.
Limit of Quantitation (LOQ)	The minimum concentration at which the concentration of an analyte may be reported. Generally taken to be the concentration that produces an analytical signal 10 times greater than the noise level.

1. Determine the S/N, LOD, and LOQ for the measurement in Table 1.D.1.

The level of the blank, given by the average of the blank measurements, is easily subtracted from any measurement. The limit to our ability to make a measurement is how much this blank varies. Thus the lowest concentration that we can determine is limited by the intrinsic noise in the measurement, which is measured by the standard deviation of the blank. Type the labels 'average' and 'std dev' below the data table in cells B23 and B24. In cells C23:D24 type the formulas:

C23: =AVERAGE(C18:C22)

C24: =STDEV(C18:C22)

D23: =AVERAGE(D18:D22)

D24: =STDEV(D18:D22)

The blank standard deviation in cell C24 provides a measure of the intrinsic noise of the measurement. For the LOD we use three times this noise level and for the LOQ we use 10 times this noise level. The labels "3x noise =", "10x noise =", and "calibration" are entered to the right of the data table. In cells H19 and H20 enter:

H19: =3*\$C\$24

H20: =10*\$C\$24

These results are the signal levels in cps for the LOD and LOQ. We can use the result from the 10.0 ppm standard as calibration to convert these values to concentration

units. Since the blank is non-zero, we subtract it from the measurement. We will also convert the units from ppm to ppb:

$$\frac{\text{measured signal}}{\text{concentration}} = \frac{1734259 \text{ cps} - 630 \text{ cps}}{(10.0 \text{ ppm})(1000 \text{ ppb/ppm})} = 173.4 \text{ cps/ppb}$$

In this case the blank measurement is small enough to neglect, but I will retain it in the calculation to be correct if I reuse the formulas. In cell H21 enter the formula:

$$=(D23-C23)/(H17*H18)$$

In cells H23 and H24 enter the formulas:

$$\text{H23: } =H19/H21$$

$$\text{H24: } = H20/H21$$

These formulas will calculate the LOD and LOQ in the units of ppb. Modify the formulas in F15 and F16 by adding "1000*" and enter the units "ppb" in cells G15 and G16. Adjust the number format if necessary and highlighting produces something like the following figure:

	A	B	C	D	E	F	G	H	I
15	Table 1.F.1. Repetitive fluorescence measurements.¹								
16			blank	standard					
17		trial	(cps)	(cps)		standard concentration:		10.0	ppm
18		1	753	1667287				1000	ppb/ppm
19		2	620	1670694		3x noise =		159	cps
20		3	703	1670408		10x noise =		531	cps
21		4	620	1668215		calibration =		173	cps/ppb
22		5	576	1927720					
23		average	630	1734259			LOD:	0.9	ppb
24		std dev	53	128979			LOQ:	3.1	ppb

- Correct the measurements in Table 1.D.2 for the nonzero background and plot. Determine the LOD and LOQ.


I left column C empty in Table 1.D.2 to make it easy to highlight the data to plot. To correct the data I find the average blank measurement. In cells D36 and D37 enter:

```
=AVERAGE(E30:E35)
=STDEV(E30:E35)
```

The average value in cell D36 is then used to correct the measured signal for the nonzero background. In cell C30 enter the formula:

```
=D30-$D$36
```

Be sure to use absolute referencing for cell D36, then copy this formula down to cell C35. Now highlight cells B30:C35 and insert a scatter chart. Add a trendline to the data and show the equation on the chart. Since we've subtracted the background to correct the data, we can justify forcing the intercept through 0,0.



In the figure to the right I've set the maximum values for the x and y axes to expand the lower concentration data. Plotting the data is not necessary for determining the LOD and LOQ, but it gives us a visual tool of what we are doing. The error bars in the plot are set to the LOD that we find below.

In cells G32:G34 enter the labels as done previously. In cells H32 and H33 enter the formulas for the 3x and 10x noise levels:

```
=3*$E$37
=10*$E$37
```

In cell H34 enter the following formula to find the slope of the line for the data:

```
=SLOPE(C30:C35,B30:B35)
```

In cells H36 and H37 enter the formulas to calculate the LOD and LOQ in units of mM:

H23: =H19/H21

H24: = H20/H21

Adding labels and formatting will give something like the following figure:

	A	B	C	D	E	F	G	H	I
27	Table 1.F.2. Fluorescence signal as a function of concentration.²								
28	conc	corrected	measured	blank					
29	(mM)	signal (mV)	signal (mV)	signal (mV)					
30	0.300	0.576	0.581	0.006					
31	0.095	0.168	0.173	0.003					
32	0.030	0.054	0.059	0.006		3x noise =	0.004		
33	0.009	0.024	0.029	0.005		10x noise =	0.013		
34	0.003	0.005	0.010	0.006		slope =	1.911		
35	0.000	0.002	0.007	0.004					
36			average	0.0050		LOD:	0.002	mM	
37			std dev	0.0013		LOQ:	0.007	mM	

What we are saying is that we can detect the analyte at 0.002 mM or higher and that we can quantitate the analyte at 0.007 mM or higher. Compare these values to the scatter in the data points in the plot above. A sample that produces a fluorescence signal between 0.004 and 0.013 we say is detectable, and a sample with a signal greater than 0.013 we will use the calibration line to report a concentration value.

1.E calibration

This worksheet contains measurement results of acetylsalicylic acid (ASA or aspirin) using light absorption. We do not need to know the details of the absorption measurement, but the key point is that a measured absorbance, A , is directly proportional to analyte concentration:

$$A \propto c_{ASA}$$

The exercise is to use the measurement of a standard solution of ASA to determine the concentration in solutions of unknown concentration.

Concepts	Concept Synopsis
proportionality	Setting up a proportionality expression to relate an unknown to a known.
calibration curve	Using multiple pairs of x,y data to construct a proportionality curve.

1. Calculate the ASA concentration in the unknown samples using a simple proportionality.

The measurement procedure is given in the worksheet for completeness. The main point is that the measured absorbance, A , is directly proportional to concentration. Given $A \propto c_{ASA}$, we can set up a proportionality expression for measurements made using identical conditions:

$$\frac{c_{\text{unk}}}{A_{\text{unk}}} = \frac{c_{\text{std}}}{A_{\text{std}}}$$

Here the 'unk' subscript refers to the solution of unknown concentration and 'std' refers to the standard solution. All values except c_{unk} are known, so we solve for c_{unk} by entering the two absorbance measurements and the standard concentration.

An equivalent approach is to write the proportionality expression for the standard:

$$A_{\text{std}} = Z c_{\text{std}}$$

where Z is the proportionality factor for the particular experimental conditions. On rearranging for Z we have:

$$Z = \frac{A_{\text{std}}}{c_{\text{std}}}$$

We can use this known proportionality factor to find c_{unk} for an unknown measurement:

$$A_{\text{unk}} = Z c_{\text{unk}}$$

Rearranging gives an expression to solve for the unknown concentration:

$$c_{\text{unk}} = \frac{A_{\text{unk}}}{Z}$$

These two approaches to solve the calculation are equivalent. Use whichever variation that you find most intuitive and the least error-prone.

The measurements and standard concentration are already entered in Table 1.E.1:

	A	B	C	D	E	F
16			Table 1.D.1			
17				absorbance	blank	conc (M)
18			standard	0.363	0.000	5.00E-05
19			Sample 1	0.222	0.000	
20			Sample 2	0.311	0.043	

Using the calculation:

$$c_{\text{unk}} = \frac{A_{\text{unk}} c_{\text{std}}}{A_{\text{std}}}$$

For Sample 1 I type in cell F19:

$$=D19*F18/D18$$

The result is 3.06E-05 M.

The sample 2 calculation is almost the same, but in this case there is some background “absorbance” due to turbidity. We correct the measured absorbance by subtracting the non-zero background, A_{blank} using the calculation:

$$c_{\text{unk}} = \frac{(A_{\text{unk}} - A_{\text{blank}}) c_{\text{std}}}{A_{\text{std}}}$$

In cell F20 I enter:

$$=(D20-E20)*F18/D18$$


and the result is 3.69E-05 M. Not subtracting the blank absorbance would give an erroneous concentration of 4.28E-05 M, a result that is more than 15 % larger than using the corrected calculation.

2. Generate a two-point calibration curve using 0,0 and the standard measurement. Calculate the ASA calculation in the unknown samples using the calibration curve..

A calibration curve is a plot of an analytical signal versus analyte concentration. Besides generating an equation that is convenient to calculate the concentration of an unknown, it shows us the linear range of the method. This particular example of plotting the standard measurement and the 0,0 origin does not illustrate this advantage, and we are using our knowledge of the measurement to assume that the relationship, $A \propto C_{ASA}$, is linear between $A = 0$ and $A = A_{std}$.

Generating a trendline or using the Excel SLOPE function are equivalent to finding Z above. The only real usefulness of the plot is as a visual tool to remind us that we are assuming a linear relationship up to A_{std} . In general we do not extrapolate a calibration curve beyond our standard concentration. For UV-vis absorption, most commercial spectrophotometers will produce linear results for absorbance between 0.1 and 1.0. In this particular case, we dilute our unknown solution to be less than the concentration of the standard.

I retype the blank and standard values in a new table and then I select the data to insert a scatter plot (see Figure below). If your plot looks different, right-click in the chart, choose "Select Data...", click the "Switch Row/Column" button, and click OK. Right click on a data point, select "Add Trendline...", check the boxes to "Display equation on chart" and "Set Intercept = 0.0", and click the Close button. After formatting the axes, you should have something like the following figure.



The slope from this linear equation is the proportionality factor relating a measured absorbance (the y value) to an unknown concentration (the x value). The calculation is the same as described above for finding the proportionality constant Z. Here I add the units of the slope:

$$C_{\text{unk}} = \frac{A_{\text{unk}}}{7260 \text{ M}^{-1}}$$

A more general formula that includes a non-zero blank is:

$$C_{\text{unk}} = \frac{(A_{\text{unk}} - A_{\text{blank}})}{7260 \text{ M}^{-1}}$$

The results for Samples 1 and 2 using these expressions will be the same as for question 1 of this worksheet. Note that plotting just two calibration points is not necessary, but it illustrates the process for when we have multiple standards to generate a calibration curve.

1.F standard-addition

This worksheet contains a table of data of an electrochemical measurement versus the concentration of a standard addition. We will determine the unknown from one addition and from the line generated by all of the data.

Concepts	Concept Synopsis
standard addition	A method of calibration in which a known amount of the pure analyte is added to the sample. The relative increase in the analytical signal provides the calibration to determine the unknown concentration.

1. Determine the unknown concentration using a proportionality calculation.

The first data point (0.0 μM , 2.5 μA) is the measurement for the unknown solution with no standard addition, c_{unk} . On adding 2.5 μM of Pb^{2+} , the current signal in microampere increases from 2.4 μA to 5.2 μA . A concentration of 2.5 μM of Pb^{2+} produces an analytical signal of (5.2 μA – 2.4 μA) or 2.8 μA . We can use this result in a simple proportionality to find the unknown concentration:

$$\frac{c_{\text{unk}}}{I_{\text{unk}}} = \frac{c_{\text{std-addn}}}{I_{\text{std-addn}} - I_{\text{unk}}}$$

where I_{unk} and $I_{\text{std-addn}}$ are the measured currents for the unknown solution and a solution with added standard, respectively. Using the results for the first standard addition gives:

$$\frac{c_{\text{unk}}}{2.4 \mu\text{A}} = \frac{2.5 \mu\text{M}}{2.8 \mu\text{A}}$$

$$c_{\text{unk}} = 2.1 \mu\text{M}$$

In cell E18 I enter the formula for this calculation:

$$=C18*\$D\$17/(\$D18-\$D\$17)$$

You can copy this formula into cells E19 and E20 to perform the calculation for the other two data points. Finding the average of the three calculations will give the result in the

following figure. I've left an extra decimal place in the results to be able to compare values.

	A	B	C	D	E	F
14			Table 1.E.1			
15			std addition	signal	c_{unk}	
16			(μM)	(μA)	(μM)	
17			0.0	2.4		
18			2.5	5.2	2.14	
19			5.0	8.2	2.07	
20			7.5	11.0	2.09	
21				average:	2.10 μM	
22				std dev:	0.04	

Using a proportionality calculation of one standard addition measurement is easy to perform, but we are making an assumption that the the measurement is linear for the concentration range of the samples. Since we use standard addition when matrix effects might affect our measurement, it makes sense to prepare multiple standard additions to check that the signal response to analyte concentration is linear. The closeness of the three results in cells E18:E20 confirms a linear dependence in this case. The next part of this worksheet shows a graphical procedure to visualize the standard addition data.

2. Determine the x-intercept for the data set, which equals the unknown concentration.

Highlight the three pairs of x,y data in Table 1.F.1 and insert a scatter chart. Add a trendline and extend the line to negative values by entering a value of 3 for Forecast Backwards in the Format Trendline dialog box. Adjust the x and y axes to show the trendline negative values. The equation of the trendline can be used to find the x-intercept by setting $y=0$ and solving for x. Rather than taking the values from the plot, I use Excel functions to find these values from the data so the calculation will be correct if the data is edited. In cells C26, C27, and C28 enter the labels "slope = ", "intercept = ", and "x-intercept = ". In cells D26, D27, and D28 enter the formulas:

```
=SLOPE(D17:D20,C17:C20)
=INTERCEPT(D17:D20,C17:C20)
=D27/D26
```

Add units in column E, highlight the result, and the result will look like the following figure. We can see easily that the measurement is linear for the concentration of the sample. The proportionality calculations and this plotting procedure give slightly different

answers, but the difference is not significant. For this type of data the result should be reported as 2.1 μM Pb.

